
Inleiding

Het gebruik van de Visual Basic functie Split

Een functie van Visual Basic die ik sinds kort graag gebruik is de functie Split. In de applicaties wordt veel gewerkt met string-manipulatie en tot voor kort werd vsAwk (een third-party ocx van VideoSoft) hiervoor gebruikt. Ook van de VB-array en een eigen functie voor string-manipulatie werd gebruik gemaakt. 

Split heeft echter wat voordelen ten opzichte van de andere methoden. Er is geen externe ocx meer die elke keer moet worden geinitialiseerd. Om een string op te delen in verschillende delen zijn er minder regels code nodig (voor een programmeur toch het eerste doel ;-)). Het teruggeven van een waarde op een bepaalde positie in de string gaat vrij simpel en weer met weinig regels code. Verder is het resultaat van Split telkens opnieuw te gebruiken. Met beide andere methoden moet telkens de functie opnieuw worden uitgevoerd. Het lijkt me sneller al heb ik daar geen echte ‘harde cijfers’ van.

Met wat voorbeelden zal het gebruik van Split duidelijk worden.

1. opdelen van een string in verschillende delen:

2. teruggeven van de waarde op een bepaalde positie in de string

Open een nieuw project en voeg een form toe (name = frmSplit). Plaats daarop een textbox (name = txtString), een listbox (name = lstString) en vijf  commandoknoppen (name = cmsAwk, cmdFunction, cmdSplit, cmdClear en cmdExit). Voeg de VideoSoft ocx toe aan het project en plaats de VideoSoftAwk op het form (name = vsAwk1).

Voeg de onderstaande code toe in de juiste plaatsen:

Option Explicit

Private aText() As String

Private Sub cmdAwk_Click()

    Dim vsAwk As VideoSoftAwk

    Dim intX As Integer

    '

    Call cmdClear_Click

    Set vsAwk = Me.vsAwk1

    vsAwk.FS = " "

    vsAwk.L = txtString.Text

    For intX = 1 To vsAwk.NF

        lstString.AddItem vsAwk.F(intX)

    Next 'intX

    Set vsAwk = Nothing

End Sub

Private Sub cmdClear_Click()

    lstString.Clear

End Sub

Private Sub cmdExit_Click()

    Call Unload(Me)

End Sub

Private Sub cmdFunction_Click()

    Dim intX As Integer

    Dim strTMP As String

    '

    Call cmdClear_Click

    intX = 1

    strTMP = " "

    Do While strTMP <> ""

        strTMP= ExtractString(txtString.Text,intX," ")

        lstString.AddItem strTMP

        intX = intX + 1

    Loop

End Sub

Private Sub cmdSplit_Click()

    Dim intX As Integer

    '

    Call cmdClear_Click

    aText = Split(txtString.Text)

    For intX = LBound(aText) To UBound(aText)

        lstString.AddItem aText(intX)

    Next 'intX

End Sub

Private Sub Form_Load()

    txtString.Text = "De kat krabt de krullen “ + _

“van de trap "

End Sub

Private Function ExtractString(ByRef strValue As String, ByRef intPos As Integer, Optional strSeperator As String = "|") As String

Dim pos As Integer

Dim intNF As Integer

Dim intX As Integer

Dim strTMP As String

'

ExtractString = strValue

intNF = 1

For intX = 1 To Len(strValue)

   If Mid(strValue, intX, 1) = strSeperator Then

        intNF = intNF + 1

   End If

Next

'

If intPos > intNF Then 

    ExtractString = ""

    Exit Function

End If

If intPos = intNF Then  'last item

    pos = InStrRev(strValue, strSeperator)

    ExtractString=Right(strValue,Len(strValue)- pos)

    Exit Function

End If

strTMP = ""

For intX = 1 To intPos

  pos = InStr(strValue, strSeperator)

  strTMP = Left(strValue, pos)

  strValue=Right(strValue,Len(strValue)-Len(strTMP))

Next

ExtractString = Left(strTMP, Len(strTMP) - 1)

End Function

Private Sub Form_Unload(Cancel As Integer)

    End

End Sub

Druk op F5 en gebruik de drie verschillende methoden. Ogenschijnlijk is de uitvoering even snel maar dit voorbeeld is alleen om het aantal gebruikte regels code te laten zien. Het gebruik van Split gebeurt met een minimaal aantals regels. Split maakt een array aan; deze blijft in het gehele project bestaan. Het ophalen van een waarde op een vaste positie in de string zal dit duidelijk maken.

Voeg drie nieuwe commandoknoppen toe (name = cmdPositieAwk, cmdPositieFunction, cmdPositieSplit). Voeg de code toe op de juiste plek.

Private Const VASTEPOSITIE As Integer = 2

Private Sub cmdPositieAwk_Click()

    Dim vsAwk As VideoSoftAwk

    '

    Set vsAwk = vsAwk1

    vsAwk.FS = " "

    vsAwk.L = txtString.Text

    lstString.AddItem CStr(VASTEPOSITIE) & _

" " & vsAwk.F(VASTEPOSITIE)

    Set vsAwk = Nothing

End Sub

Private Sub cmdPositieFunction_Click()

    lstString.AddItem CStr(VASTEPOSITIE) & " " & _

    ExtractString(txtString.Text, VASTEPOSITIE, " ")

End Sub

Private Sub cmdPositieSplit_Click()

    'array starts with index = 0!

    lstString.AddItem CStr(VASTEPOSITIE) & " " & _     

    Split(txtString.Text)(VASTEPOSITIE - 1)

End Sub

Bij het gebruik van de ocx moet deze eerst opnieuw geinitialiseerd worden. Het gebruik van de functie is trager omdat deze elke keer de hele string opnieuw doorloopt. Split maakt gebruik van een array met de index als sleutel. Een array-index begint altijd met ‘0’ vandaar de gewijzigde VASTEPOSITIE. Weer een minimaal aantal regels code om het gewenste resultaat te krijgen. 

Tenslotte heeft de functie Split nog een ‘tegenhanger’ in de fucntie Join. Deze maakt van een array weer netjes een (1) string. Het scheidingsteken kan zelf worden aangegeven. 

Al met al genoeg redenen om voortaan met Split te werken.

In het voorbeeld Split.zip is een werkend project waaruit bovenstaande procedures zijn ontleend. De VideoSoft ocx is ook aanwezig. Echter deze moet handmatig worden geregistreerd en is alleen te gebruiken in de executable (vanwege het ontbreken van de licentie in de ontwikkelomgeving).
















